NEW JERSEY GEOLOGICAL SURVEY
TECHNICAL MEMORANDUM

TIP: A Theis, Interactive Program

by

Jeffrey L. Hoffman



Department of Environmental Protection
Division of Water Resources

1984

CONTENTS

I. Introduction

TIP (a Theis, Interactive Program) is a user oriented interactive program to evaluate the effect of one or more pumping/injection wells on one or more observation wells. The user may change various parameters to quickly see the effect of adding new pumping wells, changing pumping rates, or altering aquifer characteristics.

The Theis equation is used to calculate drawdown based on time, distance from pumping wells, rates of pumping, storage, and transmissivity. The principle of superposition is used to determine drawdowns when pumping occurs at two or more wells.

TIP prompts the user for the storage coefficient, transmissivity, time, name and location of pumping and observation wells, and withdrawal or injection rates of the pumping wells. TIP then calculates drawdown at all observation wells, displays the calculated drawdown, and allows the user to change any parameter and recalculate drawdowns.

It must be emphasised that the Theis equation applies to confined, homogenous aquifers that are of constant thickness and infinite areal extent, with all wells fully penetrating the aquifer. Violating any of these assumptions diminishes the accuracy of the calculated drawdowns. However, the simplicity of the Theis equation makes its use advantageous in many situations.

II. Theory

The Theis equation is:

$$d = (Q/4\pi T) \int_u^\infty e^{-x} x^{-1} dx$$

Where

d = drawdown
Q = pumping rate
T = transmissivity
x = variable of integration
$u = r^2 s/4Tt$
r = distance from pumping well
s = storage
t = time

Let: $W(u) = \int_u^\infty e^{-x} x^{-1} dx$

Then: $d = (Q/4\pi T) W(u)$

W(u) is the well function and has been tabulated for various values of u. TIP, however, evaluates the well function by

approximating the W(u) vs. u curve. This approximation is valid for the range -174 <u< 170; values of u outside this range may cause errors.

The subroutine E1 evaluates the well function. It was copied from a program written by Cleary and Ungs. (Cleary, Robert W. and Michael J. Ungs (1978). Report No. 78-WR-15, Water Resources Program, Princeton University, Princeton, New Jersey.)

III. How to Use TIP

The user must first log onto the Department of Transportation's CMS system and gain access to the files of the Groundwater Resource Evaluation Section. (See Appendix A for the logon and access procedures.) Once the files have been accessed, simply type TIP, followed by the return key. The program will then take over.

TIP will ask for a title for this session, information on the pumping wells, information on the observation wells, and the storage coefficient, transmissivity, and time. TIP will then calculate drawdowns at all observation wells and display this information. (Note: well names may consist of up to eight characters. Also, if 2 or 3 numbers need to be entered on a single line, separate the numbers by blanks. It is important to read all questions the program asks and to answer accurately, a wrong answer may result in nonsensical results.)

TIP next asks if this run should be saved. A 'y' or 'n' will answer this question. If the run is saved, it will be written onto the file TIP OUTPUT A.

TIP then asks if there are to be any changes. If no, the session ends. If yes, TIP allows for entering new values of the storage coefficient, transmissivity, and time. TIP also enables the user to delete or add pumping and observation wells, and to change pumping rates.

After some questions TIP prints out an extra question mark. This can be ignored. If a mistake is made while entering some data hitting the BREAK key will wipe out the entire line, allowing the user to input it again in the correct form. See Appendix A for the procedure to print out any runs saved.

IV. A Sample Session

```
TIP
FI 6 DISK TIP OUTPUT A
LOAD TIP ( START NOMAP
EXECUTION BEGINS...


                        THEIS INTERACTIVE PROGRAM

  THIS PROGRAM ASSUMES THE AQUIFER IS CONFINED,
  HOMOGENEOUS, OF CONSTANT THICKNESS, AND OF
  INFINITE AREAL EXTENT.



 INPUT A TITLE, UP TO 80 CHARACTERS
EXAMPLE OF TIP

 HOW MANY PUMPING WELLS
?
2

 WHAT IS NAME OF NEXT PUMPING WELL
HOME1
 WHAT ARE ITS X,Y COORDINATES?
?
-360.   470.
 WHAT IS ITS PUMPING(+) OR INJECTION(-) RATE? (GPM)
?
10.

 WHAT IS NAME OF NEXT PUMPING WELL
FACTORY
 WHAT ARE ITS X,Y COORDINATES?
?
-100.   100.
 WHAT IS ITS PUMPING(+) OR INJECTION(-) RATE? (GPM)
?
250.

 HOW MANY OBSERVATION WELLS?
_
```

```
?
3

 WHAT IS NAME OF NEXT OBSERVATION WELL?
OB1
 WHAT ARE ITS X,Y COORDINATES?
?
-200. 200.

 WHAT IS NAME OF NEXT OBSERVATION WELL?
OB2
 WHAT ARE ITS X,Y COORDINATES?
?
-400. 351.

 WHAT IS NAME OF NEXT OBSERVATION WELL?
HOME2
 WHAT ARE ITS X,Y COORDINATES?
?
-80.  -246.


 WHAT IS THE STORAGE COEFFICIENT?
?
.4
 WHAT UNITS TO USE FOR TRANSMISSIVITY?
      1 : SQUARE FEET PER DAY
      2 : GALLONS PER DAY PER FOOT
 (TYPE NUMBER OF PROPER UNITS)
?
1
 WHAT IS THE TRANSMISSIVITY?   (SQ FT/DAY)
?
300.
 WHAT UNITS TO USE FOR TIME?
      1 : DAYS
      2 : MINUTES
 (TYPE NUMBER OF PROPER UNITS)
?
1
 AT WHAT TIME IS DRAWDOWN TO BE CALCULATED?   (DAYS)
?
5


 EXAMPLE OF TIP


     STORAGE =           0.40000E+00  (CU FT/CU FT)
     TRANSMISSIVITY =    0.30000E+03  (SQ FT/DAY)
```

```
      TIME =                    5.0000   (DAYS)

                        PUMPING WELLS
   NO.      NAME              X                Y          RATE (GPM)
    1       HOME1          -360.0            470.0          10.000
    2       FACTORY        -100.0            100.0         250.000

                     OBSERVATION WELL DRAWDOWNS
   NO.      NAME              X                Y          DRAWDOWN (FEET)
    1       OB1            -200.0            200.0           1.643
    2       OB2            -400.0            351.0           0.103
    3       HOME2           -80.0           -246.0           0.000


      SAVE THIS RUN? (Y/N)
Y



 ANY CHANGES? (Y/N)
Y


 CHANGE STORAGE? (Y/N)
N
 CHANGE TRANSMISSIVITY? (Y/N)
N
 CHANGE TIME? (Y/N)
N


 ALTER ANY PUMPING PARAMETER? (Y/N)
Y


     DO YOU WANT TO DELETE A PUMPING WELL? (Y/N)
Y
  NO.     NAME
   1      HOME1
   2      FACTORY

 WHAT IS NUMBER OF WELL TO DELETE?
?
1
 DELETE ANOTHER PUMPING WELL? (Y/N)
N


 ADD NEW PUMPING WELLS? (Y/N)
```

```
Y
 HOW MANY WELLS TO ADD?
?
1

 WHAT IS NAME OF NEXT NEW PUMPING WELL?
SCHOOL
 WHAT ARE ITS X,Y COORDINATES AND PUMPING RATE? (GPM)
?
-250. 79. 100.


 CHANGE ANY PUMPING RATES? (Y/N)
Y
 WELL NO.      NAME      PUMPING RATE (GPM)
    1   FACTORY          250.000
    2   SCHOOL           100.000
 HOW MANY RATES TO CHANGE?
?
1

 WHAT IS NUMBER OF WELL WHOSE RATE IS TO BE CHANGED?
?
1
 WHAT IS NEW PUMPING RATE? (GPM)
?
200.

                    PUMPING WELLS
  NO.    NAME              X                Y          RATE (GPM)
    1      FACTORY        -100.0           100.0        200.000
    2      SCHOOL         -250.0            79.0        100.000
 SATISFIED WITH PUMPING WELLS? (Y/N)
Y


 ALTER ANY OBSERVATION WELL PARAMETERS? (Y/N)
Y

 DELETE AN OBSERVATION WELL? (Y/N)
Y

                    OBSERVATION WELLS
  NO.    NAME              X                Y
    1      OB1            -200.0           200.0
    2      OB2            -400.0           351.0
    3      HOME2           -80.0          -246.0

 WHAT IS NUMBER OF OBSERVATION WELL TO DELETE?
?
3
```

```
 DELETE ANOTHER OBSERVATION WELL? (Y/N)
N

 ADD AN OBSERVATION WELL? (Y/N)
Y
 HOW MANY OBSERVATION WELLS TO ADD?
?
2

 WHAT IS NAME OF NEW OBSERVATION WELL?
OB3
 WHAT ARE X,Y COORDINATES OF THIS WELL?
?
-642. -5.

 WHAT IS NAME OF NEW OBSERVATION WELL?
OB4
 WHAT ARE X,Y COORDINATES OF THIS WELL?
?
37. 296.
```

OBSERVATION WELLS

NO.	NAME	X	Y
1	OB1	-200.0	200.0
2	OB2	-400.0	351.0
3	OB3	-642.0	-5.0
4	OB4	37.0	296.0

```
 SATISFIED WITH OBSERVATION WELLS? (Y/N)
Y


EXAMPLE OF TIP
```

STORAGE =	0.40000E+00	(CU FT/CU FT)
TRANSMISSIVITY =	0.30000E+03	(SQ FT/DAY)
TIME =	5.0000	(DAYS)

PUMPING WELLS

NO.	NAME	X	Y	RATE (GPM)
1	FACTORY	-100.0	100.0	200.000
2	SCHOOL	-250.0	79.0	100.000

OBSERVATION WELL DRAWDOWNS

NO.	NAME	X	Y	DRAWDOWN (FEET)
1	OB1	-200.0	200.0	2.200
2	OB2	-400.0	351.0	0.001
3	OB3	-642.0	-5.0	0.000
4	OB4	37.0	296.0	0.049

```
     SAVE THIS RUN? (Y/N)
Y



 ANY CHANGES? (Y/N)
Y


 CHANGE STORAGE? (Y/N)
N
 CHANGE TRANSMISSIVITY? (Y/N)
N
 CHANGE TIME? (Y/N)
Y
 WHAT IS NEW TIME?   (DAYS)
?
10.


 ALTER ANY PUMPING PARAMETER? (Y/N)
N


 ALTER ANY OBSERVATION WELL PARAMETERS? (Y/N)
N


 EXAMPLE OF TIP


     STORAGE =          0.40000E+00  (CU FT/CU FT)
     TRANSMISSIVITY =   0.30000E+03  (SQ FT/DAY)
     TIME =                10.0000  (DAYS)

                    PUMPING WELLS
  NO.      NAME          X              Y          RATE (GPM)
   1       FACTORY      -100.0          100.0      200.000
   2       SCHOOL       -250.0           79.0      100.000

                    OBSERVATION WELL DRAWDOWNS
  NO.      NAME          X              Y          DRAWDOWN (FEET)
   1       OB1          -200.0          200.0       6.528
   2       OB2          -400.0          351.0       0.061
   3       OB3          -642.0           -5.0       0.004
   4       OB4            37.0          296.0       0.582
```

```
     SAVE THIS RUN? (Y/N)
Y


 ANY CHANGES? (Y/N)
N


     ANY RUNS SAVED ARE IN THE FILE 'TIP OUTPUT A'
R;
```

```
T TIP OUTPUT A

1

 EXAMPLE OF TIP


     STORAGE =              0.40000E+00   (CU FT/CU FT)
     TRANSMISSIVITY =       0.30000E+03   (SQ FT/DAY)
     TIME =                      5.0000   (DAYS)

                    PUMPING WELLS
 NO.      NAME              X                Y          RATE (GPM)
  1       HOME1          -360.0            470.0          10.000
  2       FACTORY        -100.0            100.0         250.000

                    OBSERVATION WELL DRAWDOWNS
 NO.      NAME              X                Y          DRAWDOWN (FEET)
  1       OB1            -200.0            200.0           1.643
```

2	OB2	-400.0	351.0	0.103
3	HOME2	-80.0	-246.0	0.000

1

EXAMPLE OF TIP

STORAGE = 0.40000E+00 (CU FT/CU FT)
TRANSMISSIVITY = 0.30000E+03 (SQ FT/DAY)
TIME = 5.0000 (DAYS)

PUMPING WELLS

NO.	NAME	X	Y	RATE (GPM)
1	FACTORY	-100.0	100.0	200.000
2	SCHOOL	-250.0	79.0	100.000

OBSERVATION WELL DRAWDOWNS

NO.	NAME	X	Y	DRAWDOWN (FEET)
1	OB1	-200.0	200.0	2.200
2	OB2	-400.0	351.0	0.001
3	OB3	-642.0	-5.0	0.000
4	OB4	37.0	296.0	0.049

1

EXAMPLE OF TIP

STORAGE = 0.40000E+00 (CU FT/CU FT)
TRANSMISSIVITY = 0.30000E+03 (SQ FT/DAY)
TIME = 10.0000 (DAYS)

PUMPING WELLS

NO.	NAME	X	Y	RATE (GPM)
1	FACTORY	-100.0	100.0	200.000
2	SCHOOL	-250.0	79.0	100.000

OBSERVATION WELL DRAWDOWNS

NO.	NAME	X	Y	DRAWDOWN (FEET)
1	OB1	-200.0	200.0	6.528
2	OB2	-400.0	351.0	0.061
3	OB3	-642.0	-5.0	0.004
4	OB4	37.0	296.0	0.582

R;

APPENDIX A

Logon, Access and Printing Procedures

These procedures are specific to the DECWRITER hard copy terminal. Procedures for the videoterminals differ very slightly.

I. Logon Procedures

1. Turn printer on.

2. Push the TALK button on the phone and dial either 2-7816 or 2-7817. After the high pitch tone starts, push the DATA button on the phone and hang up.

3. Hit the RETURN key. The computer should respond with VM/370 ONLINE. If this doesn't happen, repeat step 2. If it does, type LOGON, followed by your computer ID. Hit the return key.

4. The computer will ask for your password. Type this in and return.

5. The computer will ask for the budget code, job number, and TSAP code. Enter these and return.

6. The computer should type some header information. Hit the return key. Logon procedure is now finished.

II. Access Procedures

1. Type LINK DEPSPSRV 191 196 RR (return)

2. Type ACC 196 B/A (return)

3. Type COPY TIP * B TIP = A

   Note: Step 3 actually copies the TIP files from the DEPSPSRV account onto your account's files. If this is not desired, neglect step 3. But if step 3 is successfully accomplished, the access procedure need not be done again.

III. Printing Procedure (to be performed after running TIP)

1. Type T TIP OUTPUT A.
   This will print the output file at the terminal.

2. Type PRINT TIP OUTPUT A.
   This will cause the file to be printed on lined computer

paper at the Department of Transportation. The output will then be routed to the appropriate location if prior arrangements have been made.

JH/5:fmm

Appendix B: Sample Logon and Access

```
▲ LOGON DEPGRWTR
  ENTER PASSWORD:
▲ ■■■■■■■■
  ENTER BUDGET CODE, JOB NO. AND TSAP CODE,
   THE FORMAT IS -
  BBB NNNNNNN C
▲ DFF 8981550 P Q1
  LOGON AT 14:56:48 GMT TUESDAY 01/19/82
  * VALID JOBNUMBER/BUDGETCODE *
  IPOE R2 CMS (PUT8105) 10/25/81
```

```
▲ LINK DEPSPSRV 191 196 RR
  ENTER READ PASSWORD:
▲ ■■■■■■■■
  R;

▲ ACC 196 B/A
  B (196) R/O
  R;
```

TOF: **Appendix C: Listing of Program**

```
C
C                        T I P
C
C     A THEIS DRAWDOWN, INTERACTIVE PROGRAM
C
C
C                  WRITTEN SEPTEMBER, 1981 BY JEFFREY L. HOFFMAN
C                    NEW JERSEY DEPARTMENT OF ENVIRONMENTAL PROTECTION
C                    DIVISION OF WATER RESOURCES
C                    BUREAU OF GROUNDWATER MANAGEMENT
C
C
C         THIS PROGRAM SOLVES THE THEIS EQUATION.
C         TIP ASSUMES THE AQUFER IS HOMOGENEOUS, CONFINED, OF CONSTANT
C            THICKNESS, AND OF INFINITE AREAL EXTENT.
C         IT ASSUMES SUPERPOSITION DESCRIBES DRAWDOWN PATTERNS CREATED
C            WHEN TWO OR MORE WELLS ARE PUMPED.
C         TIP IS USER ORIENTED AND INTERACTIVE.
C         IT ASKS THE USER FOR ALL INPUT DATA.
C
C       UNITS ON INPUT DATA:
C         STORAGE:          CUBIC FEET PER CUBIC FEET
C         TRANSMISSIVITY:   SQUARE FEET PER DAY OR GALLONS PER DAY PER FOOT
C         TIME:             DAYS OR MINUTES
C         PUMPING RATES:    GALLONS PER MINUTE
C
C
C
      REAL*8 TRUNIT(2),TMUNIT(2)
      REAL*4 OXY(50,2),PXY(50,2),DRAW(50),PRATE(25)
      REAL*8 ONAME(50),PNAME(25),TITLE(10)
      INTEGER*2 YN,Y,N,TRIND,TMIND
      COMMON /BLOCK1/ ONAME,PNAME,OXY,PXY,DRAW,PRATE
      COMMON /BLOCK2/ NP,NO
      COMMON /BLOCK3/ STORE,TRAN,TIME,TRAN$,TIME$
      COMMON /BLOCK4/ TITLE,TRUNIT,TMUNIT
      COMMON /BLOCK5/ TRIND,TMIND
      DATA Y/'Y'/,N/'N'/
C
      CALL DESCRB
      CALL READIN
      CALL COMPUT
      CALL RESULT
      GO TO 20
   10 CALL NEWPRM
      CALL PUPDAT
      CALL OUPDAT
      CALL COMPUT
```

```
      CALL RESULT
   20 WRITE (5,101)
      READ (5,800) YN
      IF (YN.EQ.Y) GO TO 10
      WRITE (5,102)
C
  101 FORMAT (////1X,'ANY CHANGES? (Y/N)')
  102 FORMAT (///5X,'ANY RUNS SAVED ARE IN THE FILE ''TIP OUTPUT A''')
  800 FORMAT (A1)
C
  999 STOP
      END
    SUBROUTINE DESCRB
C
C   DESCRIBE THE PROGRAM
C
      WRITE (5,1)
      WRITE (5,2)
      WRITE (5,3)
C
    1 FORMAT (////25X,'THEIS INTERACTIVE PROGRAM')
    2 FORMAT (//2X,'THIS PROGRAM ASSUMES THE AQUIFER IS CONFINED,',
     &/2X,'HOMOGENEOUS, OF CONSTANT THICKNESS, AND OF ',
     &/2X,'INFINITE AREAL EXTENT.')
    3 FORMAT (///)
C
  999 RETURN
      END
        SUBROUTINE READIN
C
C   READ IN DATA
C
      REAL*8 TRUNIT(2),TMUNIT(2)
      REAL*8 TR1A,TR1B,TR2A,TR2B,TM1A,TM1B,TM2A,TM2B
      REAL*4 OXY(50,2),PXY(25,2),DRAW(50),PRATE(25)
      REAL*8 ONAME(50),PNAME(25),TITLE(10)
      INTEGER*2 YN,Y,N,TRIND,TMIND
      DATA Y/'Y'/,N/'N'/
      DATA TR1A/''(SQ FT/D'/,TR1B/'AY)      '/
      DATA TR2A/''(GAL/DAY'/,TR2B/'/FT)     '/
      DATA TM1A/''(DAYS)   '/,TM1B/''       '/
      DATA TM2A/''(MINUTES'/,TM2B/'')       '/
      COMMON /BLOCK1/ ONAME,PNAME,OXY,PXY,DRAW,PRATE
      COMMON /BLOCK2/ NP,NO
      COMMON /BLOCK3/ STORE,TRAN,TIME,TRAN$,TIME$
      COMMON /BLOCK4/ TITLE,TRUNIT,TMUNIT
      COMMON /BLOCK5/ TRIND,TMIND
C
C---TITLE
      WRITE (5,104)
      READ (5,802) (TITLE(I),I=1,10)
```

```
C---PUMPING WELLS
        WRITE (5,101)
        READ (5,*) NP
        DO 10 I=1,NP
          WRITE (5,102)
          READ (5,801) PNAME(I)
          WRITE (5,103)
          READ (5,*) PXY(I,1),PXY(I,2)
          WRITE (5,107)
          READ (5,*) PRATE(I)
  10    CONTINUE
C---OBSERVATION WELLS
        WRITE (5,105)
        READ (5,*) NO
        DO 20 I=1,NO
          WRITE (5,106)
          READ (5,801) ONAME(I)
          WRITE (5,103)
          READ (5,*) OXY(I,1),OXY(I,2)
  20    CONTINUE
C---PARAMETERS
        WRITE (5,110)
        READ (5,*) STORE
 50     WRITE (5,120)
        READ (5,*) TRIND
        IF ((TRIND.LT.1).OR.(TRIND.GT.2)) GO TO 50
        IF (TRIND.EQ.1) TRUNIT(1)=TR1A
        IF (TRIND.EQ.1) TRUNIT(2)=TR1B
        IF (TRIND.EQ.2) TRUNIT(1)=TR2A
        IF (TRIND.EQ.2) TRUNIT(2)=TR2B
        WRITE (5,111) TRUNIT
        READ (5,*) TRAN$
        TRAN=TRAN$
        IF (TRIND.EQ.2) TRAN=TRAN/7.48
 60     WRITE (5,121)
        READ (5,*) TMIND
        IF ((TMIND.LT.1).OR.(TMIND.GT.2)) GO TO 60
        IF (TMIND.EQ.1) TMUNIT(1)=TM1A
        IF (TMIND.EQ.1) TMUNIT(2)=TM1B
        IF (TMIND.EQ.2) TMUNIT(1)=TM2A
        IF (TMIND.EQ.2) TMUNIT(2)=TM2B
        WRITE (5,112) TMUNIT
        READ (5,*) TIME$
        TIME=TIME$
        IF (TMIND.EQ.2) TIME=TIME/1440.
C
 101    FORMAT (/1X,'HOW MANY PUMPING WELLS')
 102    FORMAT (/1X,'WHAT IS NAME OF NEXT PUMPING WELL')
 103    FORMAT (1X,'WHAT ARE ITS X,Y COORDINATES?')
 104    FORMAT (//1X,'INPUT A TITLE, UP TO 80 CHARACTERS')
 105    FORMAT (/1X,'HOW MANY OBSERVATION WELLS?')
```

```
 106     FORMAT (/1X,'WHAT IS NAME OF NEXT OBSERVATION WELL?')
 107     FORMAT (1X,'WHAT IS ITS PUMPING(+) OR INJECTION(-) RATE? (GPM)')
 110     FORMAT (//1X,'WHAT IS THE STORAGE COEFFICIENT?')
 111     FORMAT (1X,'WHAT IS THE TRANSMISSIVITY? ',2X,2A8)
 112     FORMAT (1X,'AT WHAT TIME IS DRAWDOWN TO BE CALCULATED? ',2X,2A8)
 120     FORMAT (1X,'WHAT UNITS TO USE FOR TRANSMISSIVITY?',
     &   /5X,'1 : SQUARE FEET PER DAY',
     &   /5X,'2 : GALLONS PER DAY PER FOOT',
     &   /1X,'(TYPE NUMBER OF PROPER UNITS)')
 121     FORMAT (1X,'WHAT UNITS TO USE FOR TIME?',
     &   /5X,'1 : DAYS',
     &   /5X,'2 : MINUTES',
     &   /1X,'(TYPE NUMBER OF PROPER UNITS)')
 801     FORMAT (A8)
 802     FORMAT (10A8)
C
 999     RETURN
         END
       SUBROUTINE COMPUT
C
C         COMPUTE DRAWDOWNS ACCORDING TO THEIS FORMULA
C
       REAL*8 PNAME(25),ONAME(50),U,EU
       REAL*4 PXY(25,2),OXY(50,2),PRATE(25),DRAW(50)
       COMMON /BLOCK1/ ONAME,PNAME,OXY,PXY,DRAW,PRATE
       COMMON /BLOCK2/ NP,NO
       COMMON /BLOCK3/ STORE,TRAN,TIME
C
       PI=3.14159
       DO 10 I=1,NO
  10    DRAW(I)=0.0
       DO 100 II=1,NP
         XW=PXY(II,1)
         YW=PXY(II,2)
         QW=192.51*PRATE(II)
         DO 90 JJ=1,NO
           DIST2=(OXY(JJ,1)-XW)**2 + (OXY(JJ,2)-YW)**2
           U=DIST2*STORE/(4.*TRAN*TIME)
           CALL E1(U,EU)
           DRAW(JJ)=DRAW(JJ) + QW*EU/(4.*PI*TRAN)
   90  CONTINUE
  100  CONTINUE
C
 999   RETURN
       END
       SUBROUTINE NEWPRM
C
C    ALTER PARAMETERS
C
       REAL*8 TRUNIT(2),TMUNIT(2),TITLE(10)
       INTEGER*2 YN,Y,N,TRIND,TMIND
```

```
      DATA Y/'Y'/,N/'N'/
      COMMON /BLOCK3/ STORE,TRAN,TIME,TRAN$,TIME$
      COMMON /BLOCK4/ TITLE,TRUNIT,TMUNIT
      COMMON /BLOCK5/ TRIND,TMIND
C
C---CHANGE STORAGE
      WRITE (5,101)
      READ (5,800) YN
      IF (YN.NE.Y) GO TO 20
      WRITE (5,102)
      READ (5,*) STORE
C---CHANGE TRANSMISSIVITY
   20 WRITE (5,103)
      READ (5,800) YN
      IF (YN.NE.Y) GO TO 30
      WRITE (5,104) TRUNIT
      READ (5,*) TRAN$
      TRAN=TRAN$
      IF (TRIND.EQ.2) TRAN=TRAN/7.48
C---CHANGE TIME
   30 WRITE (5,105)
      READ (5,800) YN
      IF (YN.NE.Y) GO TO 999
      WRITE (5,106) TMUNIT
      READ (5,*) TIME$
      TIME=TIME$
      IF (TMIND.EQ.2) TIME=TIME/1440.
C
  101 FORMAT (//1X,'CHANGE STORAGE? (Y/N)')
  102 FORMAT (1X,'WHAT IS NEW STORAGE VALUE?')
  103 FORMAT (1X,'CHANGE TRANSMISSIVITY? (Y/N)')
  104 FORMAT (1X,'WHAT IS NEW TRANSMISSIVITY? ',2X,2A8)
  105 FORMAT (1X,'CHANGE TIME? (Y/N)')
  106 FORMAT (1X,'WHAT IS NEW TIME? ',2X,2A8)
  800 FORMAT (A1)
C
  999 RETURN
      END
      SUBROUTINE PUPDAT
C
C       UPDATE THE PUMPING WELLS
C
      INTEGER*2 YN,Y,N
      REAL*8 PNAME(25),ONAME(50)
      REAL*4 PXY(25,2),OXY(50,2),PRATE(25),DRAW(50)
      COMMON /BLOCK1/ ONAME,PNAME,OXY,PXY,DRAW,PRATE
      COMMON /BLOCK2/ NP,NO
      DATA Y/'Y'/,N/'N'/
C
      WRITE (5,115)
      READ (5,800) YN
```

```
      IF (YN.EQ.N) RETURN
C---DELETE PUMPING WELLS
    5 WRITE (5,100)
      READ (5,800) YN
      IF (YN.NE.Y) GO TO 15
    9 WRITE (5,803)
      WRITE (5,801) (I,PNAME(I),I=1,NP)
      WRITE (5,102)
      READ (5,*) J
      NP=NP-1
      IF (J.GT.NP) GO TO 11
      DO 10 I=J,NP
        J=I+1
        PNAME(I)=PNAME(J)
        PXY(I,1)=PXY(J,1)
        PXY(I,2)=PXY(J,2)
        PRATE(I)=PRATE(J)
   10 CONTINUE
   11 WRITE (5,103)
      READ (5,800) YN
      IF (YN.EQ.Y) GO TO 9
C---ADD NEW PUMPING WELLS
   15 WRITE (5,104)
      READ (5,800) YN
      IF (YN.NE.Y) GO TO 30
      WRITE (5,105)
      READ (5,*) NPADD
      DO 20 I=1,NPADD
        J=NP+I
        WRITE (5,106)
        READ (5,802) PNAME(J)
        WRITE (5,107)
        READ (5,*) PXY(J,1),PXY(J,2),PRATE(J)
   20 CONTINUE
      NP=NP+NPADD
C---CHANGE PUMPING RATES
   30 WRITE (5,108)
      READ (5,800) YN
      IF (YN.NE.Y) GO TO 40
      WRITE (5,804)
      WRITE (5,805) (I,PNAME(I),PRATE(I),I=1,NP)
      WRITE (5,109)
      READ (5,*) JJ
      DO 35 I=1,JJ
        WRITE (5,110)
        READ (5,*) II
        WRITE (5,111)
        READ (5,*) PRATE(II)
   35 CONTINUE
C---UPDATED PUMPING VALUES
   40 WRITE (5,806)
```

```
      WRITE (5,807) (I,PNAME(I),PXY(I,1),PXY(I,2),PRATE(I),I=1,NP)
      WRITE (5,112)
      READ (5,800) YN
      IF (YN.EQ.N) GO TO 5
C
  100 FORMAT (//1X,'DO YOU WANT TO DELETE A PUMPING WELL? (Y/N)')
  102 FORMAT (/1X,'WHAT IS NUMBER OF WELL TO DELETE?')
  103 FORMAT (1X,'DELETE ANOTHER PUMPING WELL? (Y/N)')
  104 FORMAT (//1X,'ADD NEW PUMPING WELLS? (Y/N)')
  105 FORMAT (1X,'HOW MANY WELLS TO ADD?')
  106 FORMAT (/1X,'WHAT IS NAME OF NEXT NEW PUMPING WELL?')
  107 FORMAT (1X,'WHAT ARE ITS X,Y COORDINATES AND PUMPING RATE? (GPM)')
  108 FORMAT (//1X,'CHANGE ANY PUMPING RATES? (Y/N)')
  109 FORMAT (1X,'HOW MANY RATES TO CHANGE?')
  110 FORMAT (/1X,'WHAT IS NUMBER OF WELL WHOSE RATE IS TO BE CHANGED?')
  111 FORMAT (1X,'WHAT IS NEW PUMPING RATE? (GPM)')
  112 FORMAT (1X,'SATISFIED WITH PUMPING WELLS? (Y/N)')
  115 FORMAT (//1X,'ALTER ANY PUMPING PARAMETER? (Y/N)')
  800 FORMAT (A1)
  801 FORMAT (2X,I2,5X,A8)
  802 FORMAT (A8)
  803 FORMAT (2X,'NO.',4X,'NAME')
  804 FORMAT (1X,'WELL NO.',5X,'NAME',5X,'PUMPING RATE (GPM)')
  805 FORMAT (5X,I2,6X,A8,4X,F10.3)
  806 FORMAT (/20X,'PUMPING WELLS',/2X,'NO.',3X,'NAME',14X,'X',15X,'Y',
     &10X,'RATE (GPM)')
  807 FORMAT (2X,I2,5X,A8,2X,F10.1,5X,F10.1,5X,F10.3)
C
  999 RETURN
      END
      SUBROUTINE OUPDAT
C
C     UPDATE THE OBSERVATION WELLS
C
      INTEGER*2 YN,Y,N
      REAL*8 ONAME(50),PNAME(25)
      REAL*4 OXY(50,2),PXY(25,2),DRAW(50),PRATE(25)
      DATA Y/'Y'/,N/'N'/
      COMMON /BLOCK1/ ONAME,PNAME,OXY,PXY,DRAW,PRATE
      COMMON /BLOCK2/ NP,NO
C
      WRITE (5,101)
      READ (5,800) YN
      IF (YN.EQ.N) RETURN
C---DELETE OBSERVATION WELLS
   10 WRITE (5,102)
      READ (5,800) YN
   19 IF (YN.NE.Y) GO TO 30
      WRITE (5,805)
      WRITE (5,806) (I,ONAME(I),OXY(I,1),OXY(I,2),I=1,NO)
      WRITE (5,103)
```

```
      READ (5,*) NDEL
      NO=NO-1
      IF (NDEL.GT.NO) GO TO 25
      DO 20 I=NDEL,NO
        J=I+1
        ONAME(I)=ONAME(J)
        OXY(I,1)=OXY(J,1)
        OXY(I,2)=OXY(J,2)
   20 CONTINUE
   25 WRITE (5,104)
      READ (5,800) YN
      IF (YN.EQ.Y) GO TO 19
C---ADD OBSERVATION WELLS
   30 WRITE (5,105)
      READ (5,800) YN
      IF (YN.NE.Y) GO TO 40
      WRITE (5,106)
      READ (5,*) NADD
      DO 35 I=1,NADD
        J=NO+I
        WRITE (5,107)
        READ (5,810) ONAME(J)
        WRITE (5,108)
        READ (5,*) OXY(J,1),OXY(J,2)
   35 CONTINUE
      NO=NO+NADD
C---UPDATED OBSERVATION WELLS
   40 WRITE (5,805)
      WRITE (5,806) (I,ONAME(I),OXY(I,1),OXY(I,2),I=1,NO)
      WRITE (5,109)
      READ (5,800) YN
      IF (YN.EQ.N) GO TO 10
C
  101 FORMAT (//1X,'ALTER ANY OBSERVATION WELL PARAMETERS? (Y/N)')
  102 FORMAT (/1X,'DELETE AN OBSERVATION WELL? (Y/N)')
  103 FORMAT (/1X,'WHAT IS NUMBER OF OBSERVATION WELL TO DELETE?')
  104 FORMAT (1X,'DELETE ANOTHER OBSERVATION WELL? (Y/N)')
  105 FORMAT (/1X,'ADD AN OBSERVATION WELL? (Y/N)')
  106 FORMAT (1X,'HOW MANY OBSERVATION WELLS TO ADD?')
  107 FORMAT (/1X,'WHAT IS NAME OF NEW OBSERVATION WELL?')
  108 FORMAT (1X,'WHAT ARE X,Y COORDINATES OF THIS WELL?')
  109 FORMAT (/1X,'SATISFIED WITH OBSERVATION WELLS? (Y/N)')
  800 FORMAT (A1)
  805 FORMAT (/20X,'OBSERVATION WELLS',/2X,'NO.',3X,'NAME',14X,'X',15X,'
     &Y')
  806 FORMAT (2X,I2,5X,A8,2X,F10.1,5X,F10.1)
  810 FORMAT (A8)
C
  999 RETURN
      END
      SUBROUTINE RESULT
```

```
C
C  PRINT OUT RESULTS
C
      REAL*8 TRUNIT(2),TMUNIT(2)
      REAL*8 ONAME(50),PNAME(25),TITLE(10)
      REAL*4 OXY(50,2),PXY(25,2),DRAW(50),PRATE(25)
      COMMON /BLOCK1/ ONAME,PNAME,OXY,PXY,DRAW,PRATE
      COMMON /BLOCK2/ NP,NO
      COMMON /BLOCK3/ STORE,TRAN,TIME,TRAN$,TIME$
      COMMON /BLOCK4/ TITLE,TRUNIT,TMUNIT
      INTEGER*2 YN,Y,N
      DATA Y/'Y'/,N/'N'/
C
C---WRITE RESULTS ON SCREEN
      WRITE (5,801) (TITLE(I),I=1,10)
      WRITE (5,101) STORE,TRAN$,TRUNIT,TIME$,TMUNIT
      WRITE (5,102)
      WRITE (5,103) (I,PNAME(I),PXY(I,1),PXY(I,2),PRATE(I),I=1,NP)
      WRITE (5,104)
      WRITE (5,103) (I,ONAME(I),OXY(I,1),OXY(I,2),DRAW(I),I=1,NO)
C---HARDCOPY RESULTS
      WRITE (5,105)
      READ (5,800) YN
      IF (YN.NE.Y) GO TO 999
      WRITE (6,100)
      WRITE (6,801) (TITLE(I),I=1,10)
      WRITE (6,101) STORE,TRAN$,TRUNIT,TIME$,TMUNIT
      WRITE (6,102)
      WRITE (6,103) (I,PNAME(I),PXY(I,1),PXY(I,2),PRATE(I),I=1,NP)
      WRITE (6,104)
      WRITE (6,103) (I,ONAME(I),OXY(I,1),OXY(I,2),DRAW(I),I=1,NO)
C
 100  FORMAT (T1,'1')
 101  FORMAT (//
     &/5X,'STORAGE =          ',E12.5,2X,'(CU FT/CU FT)',
     &/5X,'TRANSMISSIVITY =   ',E12.5,2X,2A8,
     &/5X,'TIME =             ',F12.4,2X,2A8)
 102  FORMAT (/20X,'PUMPING WELLS',/2X,'NO.',5X,'NAME',13X,'X',14X,'Y',1
     &0X,'RATE (GPM)')
 103  FORMAT (2X,I2,5X,A8,2X,F10.1,5X,F10.1,5X,F10.3)
 104  FORMAT (/20X,'OBSERVATION WELL DRAWDOWNS',/2X,'NO.',5X,'NAME',13X,
     &'X',14X,'Y',10X,'DRAWDOWN (FEET)')
 105  FORMAT (///5X,'SAVE THIS RUN? (Y/N)')
 800  FORMAT (A1)
 801  FORMAT (//1X,10A8)
C
 999  RETURN
      END
      SUBROUTINE E1(Y,Z)
C                              SUBROUTINE E1
C
```

```
C
C         PURPOSE:  COMPUTE THE EXPONENTIAL INTEGRAL
C
C            THIS ROUTINE IS ACCURATE TO THE EIGHTH PLACE
C
C
C       DESCRIPTION OF PARAMETERS:
C             Y    - ARGUMENT OF EXPONENTIAL INTEGRAL
C             Z    - RESULT VALUE
C             AUX  - RESULTANT AUXILIARY VALUE
C
C
C         REMARKS:
C               Y GT 170 (LT -174) MAY CAUSE UNDERFLOW (OVERFLOW)
C                FOR Y=0, THE RESULT VALUE IS SET TO -1.0E75
C
C           THIS REQUIRES NO OTHER ROUTINE
C
      IMPLICIT REAL*8 (A-H,O-Z)
      Z=0.
      IF (Y-1.) 2,1,1
 1    CONTINUE
      S=1.D0/Y
      AUX=1.D0 -S
     &*(((S+3.377358D0)*S+2.052156D0)*S+2.709479D-1)/((((S*
     &1.072553D0+5.716943D0)*S+6.945239D0)*S+2.593888D0)*S+2.709496D-1)
      XY=-Y
      IF (XY.GE.-160) Z=AUX*S*DEXP(XY)
      RETURN
 2    IF (Y+3.) 6,6,3
 3    AUX=((((((7.122452D-7*Y-1.766345D-6)*Y+2.928433D-5)*Y-2.335379D-4
     &)*Y+1.664156D-3)*Y-1.041576D-2)*Y+5.555682D-2)*Y-2.500001D-1)*Y
     &+9.999999D-1
      Z=-1.D75
      IF (Y) 4,5,4
 4    CONTINUE
      Z=(Y*AUX-DLOG(DABS(Y))-5.772157D-01)
 5    RETURN
 6    IF (Y+9.) 8,8,7
 7    CONTINUE
      AUX=1.D0
     &-((((5.176245D-2*Y+3.061037D0)*Y+3.243665D1)*Y+2.244234D2)*Y
     &+2.486697D2)/((((Y+3.995161D0)*Y+3.893944D1)*Y+2.263818D1)*Y
     &+1.807837D2)
      GO TO 9
 8    CONTINUE
      S=9.D0/Y
      AUX=1.D0-S*(((S+7.659824D-1)*S-7.271015D-1)*S-1.080693D0)/((((S
     &*2.518750D0+1.122927D1)*S+5.921405D0)*S-8.666702D0)*S-9.724216D0)
 9    CONTINUE
      XY=-Y
```

```
      IF (XY.GE.-160) Z=AUX*DEXP(XY)/Y
      RETURN
      END
EOF:
```